
Symbolic Model-checking

• An effective technology for Model-Checking 

• Binary Decision diagrams (BDD Bryant’85)
• Based on Boolean logic 

• Originally designed for hardware systems

• Many BDD extensions proposed:
• Integer valued : MDD, DDD

• Multi-Terminal : ADD, MTBDD

• Hierarchical Set Decision Diagrams (SDD)
• Direct encoding of a structured specification

• High-level transition encoding with inductive 
homomorphisms

Hierarchical
Set Decision Diagram (SDD ) [FORTE’05]

• BDD : booleans, MDD : 
integers…

• Idea : hierarchy
• Label the arcs with a SET = 

Set Decision Diagram

BDD/MDD

1

SDD +
BDD/MDD or SDD = 
referenced values

• Increases sharing
• Memory gain 

• Time gain 

• cache

• traversals

With 

hierarchy

SDD offer Automatic Saturation
[ICATPN’08, TACAS’09]

• Symbolic model-checking based on transitive 
closures over the transition relation
• Dominates overall complexity 

• [BCMDH’92] based on BFS iterations

• [Roig’95] Chaining may converge faster, not 
strict BFS

• [CLS’01-CMS’03] Saturation is empirically 1 to 
3 orders of magnitude better, adapt firing 
order to DD structure

[ SDD and Automatic Saturation – ICATPN’08 ]
Defines rewriting rules to automatically and transp arently
enable saturation in SDD

Symbolic analysis of expressions

• Symbolic data structures: BDD, MDD

• k boolean variables => 2k

• Encode transitions with sets : 2k x 2k

• Use the support (only k’ vars) of transitions

• Build clusters of transitions

• Reorder evaluations in fixpoint computation 
(chaining, saturation)

• But :

• Exponential worst case complexity when k’ grows

• In general, necessary to invoke an explicit solver for 
each new state in 2k’

DDD support high level transition 
relations (NEW CAV’2013)

• Computing the support

• A[x + y]  => pessimistic assumptions

• x=x+1; y=y+1 => Atomic sequence of 
updates produce artificially large support

• What if we could compute this on the fly ?

• Carry the expression in a dedicated operation

• Traverse a state -> path

• Resolve variables as they are encountered 

• Drop pessimistic assumptions ASAP

• But we must still reason with sets ! 



7GAL Y. Thierry-Mieg, Sept. 2013, 68NQRT

An equivalence relation ?

• Partial expression evaluation

• f = a + b

• States : s1:(a=1, b=0)     s2:(a=0,b=1)

• If both a and b are known : f(s1)=f(s2), 
s1 and s2 are equivalent

• If only a is known, f(s1)=1+b    f(s2)=0+b
s1 and s2 are NOT equivalent

• Algorithm discovers variable values and builds
equivalence classes on the fly

Evaluating expressions over DD

Performance evaluation (BEEM)



10GAL Y. Thierry-Mieg, Sept. 2013, 68NQRT

DDD/SDD Symbolic engine

• Superior performances

• Historically (2002+) : hard coded support for 
Petri nets

• Defining transition relation symbolically is
difficult => high expertise

• How to define a system to take advantage of 
symbolic engine ?

Model Driven Development
and Model-checking

• In MDD approaches

• Build a Domain Specific Language

• Use model transformation for specific
targets

• Choosing a target formalism

• Expressive enough to capture your semantics

• Efficient solution engine

• We propose ITS/GAL formalism

• Allows to express discrete state semantics

• Symbolic model-checking

Guarded Action Language

• GAL : a « Universal Semantic Assembly Language »

• Simple to use, easy C-like syntax

• Straightforward Petri net style concurrent semantics

• Integer variables and arrays + arbitrarily nested
array expressions

• Efficient symbolic solution engine

• Subsumed by Instantiable Transition Systems (ITS), 
allowing hierarchical composition of GAL modules

• Meant to be a back-end target in a transformation 
process. 

• Define your semantics in GAL.

GAL example

GAL system { 

// Variable declarations

int variable = 5 ;

array [2] tab = (1, 2) ; 

transition t1 [variable > 9] {

tab [0] = tab [1] * tab [0] ;

variable = variable * 5 ;

}

transition t2 [variable == 23] label "a" {

tab [1] = 0 ;

}

}

GAL Variables

• All variables are 32 bit integers or arrays of 
fixed size.

• Any variable must be initialized

• int a = 0;   

• array [3] tab = (0,0,0); 

Arithmetic and Boolean

• Terminal expressions are signed constants, 
parameters, variables, array access with 
arbitrary index expression

• 3,-2,$MAX,x,tab[x+1],tab[tab[$MAX-x]]

• All C operators supported

• Bitwise : &, |, ^, <<, >>, ~

• Integer : +, -, *, /, %, **

• Boolean expressions

• Basics : true, false, &&, ||, !

• Comparisons of integers: ==,!=, <,<=, >,>=

• x = (y == 255) * 100;   // x is 0 or 100

GAL Statements

• <lhs=rhs> assign integer expression rhs to 
variable designated by lhs.

• <s1;..;sn> sequence of statements, <nop> the 
empty sequence

• <ite(c,t,f)> an if-then-else statement

• <abort> return the empty set (!)

• <for(min, max, b)> a limited form of iteration 

• <fixpoint(b)> fixpoint statement

• <call(a)> call a label (i.e. an arbitrary 
transition with label a) of « self » 

GAL Transitions

• Tuple : <label, guard, body>

• Fire : In any state where guard is enabled, 
process body statement(s) atomically

• Tau (empty) label for local transitions

• Labeled transitions are not fireable by Locals 
outside of call or synchronization

System Parameters

• Allow easier configuration of a model

GAL paramSystem ($N = 2, $K = 1) {

int variable = $N ;

array [2] tab = ($N + $K, $N - 1) ;

transition t1 [variable > $N] {

tab [$K] = tab [1] * tab [0] ;

variable = variable * 5 ;

}

transition t2 [variable == $N] label "a" {

tab [1] = 0 ;

}

}

Range Type definitions & 
Transition parameters 

GAL paramDef ($N=2) {

typedef paramType = 0..$N;

typedef paramType2 = 0..1;

int variable = 0;

// a transition compactly modeling ($N+1)*2 

// basic transitions

transition trans (paramType $p1, paramType2 $p2) 
[$p1 != $p2] {

variable = $p1 + $p2;

}

}

If then else

GAL iteExample {

int variable = 0 ;

transition invert [variable == 0 || variable == 1] {

if (variable == 0) {

variable = 1;

} else {

variable = 0;

}

}

}
Equivalent to xor : variable = variable ^ 1

For loop

• Limited iteration

GAL forLoop {
typedef Dom = 0..2;

array [3] tab = (0,0,0);

transition forExample [true] {
for ($i : Dom) {

tab[$i] = $i;
}

}
}

GAL forLoop_inst {

array [3] tab = (0, 0, 0) ;

transition forExample [true] {

tab [0] = 0 ;

tab [1] = 1 ;

tab [2] = 2 ;

}

}

Call

GAL callExample {

int variable = 0 ;

transition NDassignX [variable == 0 || variable == 1] {

self."setX" ;

}

transition callee1 [true] label "setX" {

variable = 1 ;

}

transition callee2 [true] label "setX" {

variable = 0 ;

}

}

Abort

GAL abortExample ($EFT = 1, $LFT = 3) {

int a = 1 ;

int b = 0 ;

int t.clock = 0 ;

transition t [a >= 1 && t.clock >= $EFT] {

a = a - 1 ;

b = b + 1 ;

t.clock = 0 ;

}

Abort (2)

transition elapse [true] label "elapse" {

// is t enabled ?

if (a >= 1) {

// is t's clock strictly less than 

// its latest firing time ?

if (t.clock < $LFT) {

// if yes increment t clock

t.clock = t.clock + 1 ;

} else {

// otherwise, time cannot elapse, 

//  kill exploration

abort ;

}

Fixpoint : sort example

GAL sortEx {

typedef index = 0..3;                       // 0 to  n-2

array [5] tab = (3,1,2,4,5);

int tmp = 0;

transition swap (index $i) [ tab[$i] > tab[$i+1] ]  label "sort"

{ tmp = tab[$i];  tab[$i] = tab[$i+1]; tab[$i+1] = tmp; tmp = 0;   }

transition sorted label "sort" [true] {

for ($i : index) {

if (tab[$i] > tab[$i+1]) { abort; } } }

transition sort [true] {

fixpoint { self."sort"; }

}

}

Transient

GAL loopTransient { 

int i = 0 ;

array [4] tab = (0, 0, 0, 0) ;

transition t1 [i < 4] {

tab [i] = i ;

if (i < 3) {

i = i + 1 ;

} else {

i = 0 ;

}

}

TRANSIENT = (i != 0) ;

}

Some applications of GAL

• Petri nets

• Discrete Time Petri nets

• Colored Petri Nets

• Divine (Promela-like) models

• CCSL clock logic

• …

Encoding Petri nets

• Each place => a variable

• Each transition => a transition

• Guard tests enabling conditions

• Actions update state variables

• Easy to support many extensions of PN

• Test arcs

• Reset arcs

• Inhibitor

• Capacity places

• …

Labeled Discrete TPN model of a train

Discrete Time Petri Net Semantics

• Place -> integer variable, 

• initial value=initial marking

• Transitions -> define variable t.clock

• unless [0,0] or [0,inf[

• Time elapse -> additional transition labeled « elapse »

• Sequence, for each transition t

If (enabled(t)) {

If ( t.clock < lft(t) ) {

t.clock=t.clock+1;

} else {

abort;

}

}

General case

If (enabled(t)) {

abort;

}

[0,0] urgent case

If (enabled(t)) {

If ( t.clock < eft(t) ) {

t.clock=t.clock+1;

}

}

[a,inf[ infinite lft case

DTPN : Transitions

• Transition t -> transition <l,g,b>

• l: Label is copied from input transition, 

• g: enabled(t) and t.clock >= eft(t)

• b: 
• update place markings according to arc types and inscriptions, 

• then reset current clock, 

• then reset any disabled transition clocks (call(reset))

• Reset disabled transitions: <reset,private,true,b> for 
each transition t :

If (! enabled(t)) {

t.clock=0;

}

General case If no clock do nothing

Train Example, GAL

GAL train { 

 int far = 1 ; 

 int close = 0 ; 

 int on = 0 ; 

 int left = 0 ; 

 int enter.clock = 0 ; 

 int leave.clock = 0 ; 

 

 transition SendApp [ far >= 1 ] label "SendApp" { 

  far = far - 1 ; 

  close = close + 1 ; 
  self.reset ; 

 } 

 

 transition enter [ close >= 1 && enter.clock >= 3 ] { 

  close = close - 1 ; 

  on = on + 1 ; 

  enter.clock = 0 ; 

  self.reset ; 

 } 

 

 transition leave [ on >= 1 && leave.clock >= 2 ] { 

  on = on - 1 ; 
  left = left + 1 ; 

  leave.clock = 0 ; 

  self.reset ; 

 } 

 

 transition SendExit [ left >= 1 ] label "SendExit" { 

  left = left - 1 ; 

  far = far + 1 ; 

  self.reset ;   

 } 

 

 
 

 

 

 

 

 transition elapse [ True ] label "elapse" { 

  if (close >= 1) { 

   if (enter.clock < 5) { 

    enter.clock = enter.clock + 1 ; 

   } else { 

    abort ; 

   } 

  } 

  if (on >= 1) { 
   if (leave.clock < 4) { 

    leave.clock = leave.clock + 1 ; 

   } else { 

    abort ; 

   } 

  } 

  if (left >= 1) { 

   abort ; 

  } 

 } 

 

 transition reset [ True ] label "reset" { 
  if (! close >= 1) { 

   enter.clock = 0 ; 

  } 

  if (! on >= 1) { 

   leave.clock = 0 ; 

  } 

 } 

  

} 

transition enter [ close >= 1 && enter.clock >= 3 ] {

close = close - 1 ;

on = on + 1 ;

enter.clock = 0 ;

self.reset ;

}

Using a fixpoint

• Essential states construction [Popova]

• Letting time elapse cannot disable transitions

• Let time progress, only consider states that
immediately follow a discrete transition

Time Elapse

Discrete Transition

Essential State

Compute green set : 

let time elapse if it can

cumulate states

Fire transitions (succ) from resulting set

Fixpoint for « Essential States » of TPN 

GAL tpnModel ($EFT = 3, $LFT = 5) {

int a = 1 ;

int b = 0 ;

int t.clock = 0 ;

transition t [a >= 1 && t.clock >= $EFT] label "succ"      

{

a = a - 1 ;

b = b + 1 ;

t.clock = 0 ;

}

Essential States (2)

transition elapseIfpossible

[ a >= 1 && t.clock < $LFT] label "elapseIP" {

t.clock = t.clock + 1 ;

}

transition id [true] label "elapseIP" {

}

transition nextState [true] {

fixpoint {

self."elapseIP" ;

}

self."succ" ; // Any discrete transition

}

}

Colored Petri nets

• Each place => an array of dimension 
proportional to domain

• Uncolored places => size 1

• Formal parameters => typedef of a range

• Each transition => transition with parameters

An example

CPN example

GAL DrinkVending2 {

typedef Options = 0 .. 1 ;

typedef Products = 0 .. 1 ;

typedef Quality = 0 .. 7 ;

array [8] ready = (0, 0, 0, 0, 0, 0, 0, 0) ;

array [8] wait = (1, 1, 1, 1, 1, 1, 1, 1) ;

array [2] theProducts = (1, 1) ;

array [2] productSlots = (0, 0) ;

array [2] theOptions = (1, 1) ;

array [2] optionSlots = (0, 0) ;

Color Domains

M=8, N=2

Places

CPN example (2)

transition serve (Quality $x) [ready [$x] >= 1] {

ready [$x] = ready [$x] - 1 ;

wait [$x] = wait [$x] + 1 ;

}



40GAL Y. Thierry-Mieg, Sept. 2013, 68NQRT

CPN example (3)

transition elaborate2(Options $o1,Products $p,Quality $x,Options $o2) 
[$x > 3 && $x <= 5 && theProducts [$p] >= 1

&& theOptions [$o1] >= 1 && theOptions [$o2] >= 1 

&& wait [$x] >= 1] {

theProducts [$p] = theProducts [$p] - 1 ;

theOptions [$o1] = theOptions [$o1] - 1 ;

theOptions [$o2] = theOptions [$o2] - 1 ;

wait [$x] = wait [$x] - 1 ;

optionSlots [$o2] = optionSlots [$o2] + 1 ;

optionSlots [$o1] = optionSlots [$o1] + 1 ;

productSlots [$p] = productSlots [$p] + 1 ;

ready [$x] = ready [$x] + 1 ;

}

Divine : a language for concurrent 
process

byte id;

byte t[3] = { 255 ,255 ,255  };

process P_0 {

state NCS, try, wait, CS;

init NCS;

trans 

NCS -> try 

{ guard id == 0; effect t[0] = 2;}, 

try -> wait 

{ effect t[0] = 3, id =0 +1; }, 

wait -> wait 

{ guard t[0] == 0; effect t[0] = 255;}, …

Global Variables

Process + Channels

Analyzing Divine models

• Translate Divine concepts to GAL

• Process state => variable

• Divine variables and arrays => GAL equivalent

• Guards, Instructions => GAL equivalent

• Synchronizations => use GAL call semantics

• Channels => GAL arrays + variable for size

Composite Model

• A composite contains a set of ITS instances

• State is cartesian product of subcomponent
states

• Synchronizations : <label,body>

• <s1;..;sn> sequence of statements

• <i.call(a)> call a label of subcomponent i.

• <self.call(a)> call a label (i.e. an arbitrary
synchronization with label a) of « this »

Composite example

A GAL counter

A controlled gate

Performance: Reachability

Conclusion

• GAL modeling

• A natural model for many discrete semantics

• Efficient symbolic solution 

• ITS Composite for compositional modeling

• Modular and hierarchical specifications

• Efficient support for symmetric models

• Model checking engine

• Reachability (shortest traces) 

• CTL (Forward algorithms, traces)

• LTL with Spot (Fully symbolic or hybrid)

GAL Eclipse plugin (Thanks Xtext !)

Thank you for your attention !

SDD and ITS-tools are distributed as an 
open-source LGPL/GPL C++ source and 
pre-compiled tools : 

http://ddd.lip6.fr
Eclipse plugin for GAL/ITS manipulation 
and CTL model-checking

http://coloane.lip6.fr/night-updates


